Computer und Software 1

Christof Köhler

4. Maple – Differenzieren, Integrieren, Taylorreihen

Folien: Balint Aradi

Differenzieren eines Ausdrucks

- Maple Ausdrücke werden mit diff() differenziert:

```
> diff(sin(x), x);
```

$$\cos(x)$$

Variable, nach der abgeleitet werden soll

Ausdruck, der abgeleitet werden soll

- Höhere Ableitungen werden durch Anfügen von '$n' an die Variable erzeugt.

```
> diff(x^3, x$2);
```

$$6\,x$$

Zweite Ableitung

```
> diff(1/x, x$3);
```

$$-\frac{6}{x^4}$$

Dritte Ableitung

- Das Ergebnis kann mit den üblichen Funktionen vereinfacht bzw. weiter umgeformt werden:

```
> diff(sqrt(1+x^2), x$3);
  simplify(%);
```

$$\frac{3\,x^3}{(1+x^2)^{(5/2)}} - \frac{3\,x}{(1+x^2)^{(3/2)}}$$

$$-\frac{3\,x}{(1+x^2)^{(5/2)}}$$

Träge Differentiation

- Differentiation in träger Form **Diff()** zeigt nur die Operation an.
Ausgewertet wird mittels **value()**:

```
> Diff(x^3, x$2);
  value(%);
```

$$\frac{d^2}{dx^2}(x^3)$$

$$6\,x$$

Sollte immer wenn möglich angewendet werden, damit man sieht, was genau differenziert wird.

- Die Verwendung des **%**-Operators ist nur in der zweiten (oder in einer nachfolgenden) Anweisung einer Ausführungsgruppe sicher, ansonsten **hängt das Ergebnis von der Ausführungsreihenfolge** der einzelnen Gruppen ab!

sicher

```
> Diff(x^3, x$2);
  value(%);
```

$$\frac{d^2}{dx^2}(x^3)$$

$$6\,x$$

unsicher!

```
> Diff(x^3, x$2);
```

$$\frac{d^2}{dx^2}(x^3)$$

```
> value(%);
```

$$6\,x$$

Differentiation von Funktionen

- Funktionen werden mit dem Operator **D()** abgeleitet:

```
> f := x -> 1/(sqrt(1+x^2));
```

$$f := x \to \frac{1}{\sqrt{1+x^2}}$$

```
> d1f := D(f);
```

$$d1f := x \to -\frac{x}{\sqrt{1+x^2}^3}$$

Das **Ergebnis** ist selber auch **eine Funktion**

- Höhere Ableitungen werden durch Angabe der Variable und des Ableitungsgrades erzeugt:

```
> d3f := D[1$3](f);
```

$$d3f := x \to -\frac{15\,x^3}{\sqrt{1+x^2}^7} + \frac{9\,x}{\sqrt{1+x^2}^5}$$

Funktionsvariable
(1. Variable der Funktion)

Grad der Ableitung

Differentiation von Funktionen (#2)

- Höhere Ableitungen können auch mit dem Wiederholungsoperator @@ erzeugt werden:

```
> f := x -> x^5;
```

$$f := x \rightarrow x^5$$

```
> f;
```

$$f$$

```
> f(x);
```

$$x^5$$

Funktionsdefinition wird nur dann angezeigt, wenn eine Variable eingesetzt wird und daruch ein Ausdruck erzeugt wird.

```
> d3f := (D@@3)(f);
```

$$d3f := x \rightarrow 60\ x^2$$

Der wiederholte Operator muss zwischen Klammern stehen!

- Die beiden Formen lassen sich auch mischen (allerdings äußerst unübersichtlich!):

```
> d4f := (D[1$2]@@2)(f);
```

Vierte Ableitung von f(x)

$$d4f := x \rightarrow 120\ x$$

apply/unapply

- Im Gegensatz zu Ausdrücken, lassen sich (abgeleitete) Funktionen nicht vereinfachen!

```
> d1f := simplify(D[1$3](f));
```

$$d1f := x \rightarrow -\frac{15\,x^3}{\sqrt{1+x^2}^{\,7}} + \frac{9\,x}{\sqrt{1+x^2}^{\,5}}$$

- Mit **apply()** kann eine Funktion in einen Ausdruck umgewandelt werden, mit **unapply()** kann ein Ausdruck in eine Funktion umgewandelt werden:

```
> expr := apply(d1f, x);
```

Funktion

einzusetzende Variable

$$expr := -\frac{15\,x^3}{(1+x^2)^{(7/2)}} + \frac{9\,x}{(1+x^2)^{(5/2)}}$$

```
> expr_simp := simplify(expr);
```

$$expr\_simp := -\frac{3\,x\,(2\,x^2-3)}{(1+x^2)^{(7/2)}}$$

Ausdruck

```
> d1f_simp := unapply(expr_simp, x);
```

Funktionsvariable

$$d1f\_simp := x \rightarrow -\frac{3\,x\,(2\,x^2-3)}{(1+x^2)^{(7/2)}}$$

Differentiation stückweise definierter Funktionen

- Maple kann auch stückweise definierte Funktionen ableiten:

```
> f := piecewise(x > 0, x, x <= 0, 0);
```

$$f := \begin{cases} x & 0 < x \\ 0 & x \le 0 \end{cases}$$

```
> plot(f, x=-2..2, scaling=constrained,
  thickness=3);
```




```
> diff(f, x);
```

$$\begin{cases} 0 & x < 0 \\ \mathit{undefined} & x = 0 \\ 1 & 0 < x \end{cases}$$

Ableitung in x=0 nicht definiert!

Implizites Differenzieren

- Beim Differenzieren betrachtet Maple alle Unbekannten bis auf die spezifizierte Variable als Konstanten.
- Das kann bei einer impliziten Ableitung Probleme bereiten:

```
> eq := x^2 + y^2 = 1;
```

$$eq := x^2 + y^2 = 1$$

```
> implicitplot(eq, x=-1..1, y=-1..1,
  thickness=3);
```




```
> deq := diff(eq, x);
```

$$deq := 2\,x = 0$$

Ergebnis **falsch**, da **y** als eine **Konstante** behandelt wurde!

Implizites Differenzieren (#2)

- Bei impliziter Ableitung muss die Funktionsabhängigkeit extra spezifiziert werden:

```
> eq2 := x^2 + y(x)^2 = 1;
```

$$eq2 := x^2 + y(x)^2 = 1$$

```
> deq2 := diff(eq2, x);
```

$$deq2 := 2\,x + 2\,y(x)\left(\frac{d}{dx}\,y(x)\right) = 0$$

y soll als eine Funktion von x behandelt werden

- Die implizite Ableitung kann mit **solve()** nach der d/dx(y(x)) aufgelöst werden
  Die Ableitung von y muss als **diff(y(x), x)** angegeben werden.

```
> dy := solve(deq2, diff(y(x), x));
```

$$dy := -\frac{x}{y(x)}$$

- Alternativ kann auch **implicitdiff()** auf den expliziten Ausdruck angewendet werden:

```
> implicitdiff(x^2+y^2=1, y, x);
```

$$-\frac{x}{y}$$

Funktion

Funktionsvariable

Partielles Differenzieren

- Ausdrücke mit mehreren Variablen können nach beliebiger Variable abgeleitet werden. Die Variable(n) wird/werden als Argumente von **diff()** übergeben:

```
> f := cos(x * y) / y;
```

$$f := \frac{\cos(x\,y)}{y}$$

```
> diff(f, x);
```

$$-\sin(x\,y)$$

Ableitung nach *x*: $\frac{\partial f}{\partial x}$

```
> diff(f, y);
```

$$-\frac{\sin(x\,y)\,x}{y} - \frac{\cos(x\,y)}{y^2}$$

Ableitung nach *y*: $\frac{\partial f}{\partial x}$

```
> diff(f, x, y);
```

$$-\cos(x\,y)\,x$$

Ableitung nach *x* und *y*: $\frac{\partial^2 f}{\partial x \partial y}$

- Mehrfache partielle Ableitungen werden durch das Anhängen von **$n** an den Variablennamen gebildet:

```
> diff(f, y, x$2);
```

$$\sin(x\,y)\,x\,y - \cos(x\,y)$$

$$\frac{\partial^3 f}{\partial y \partial x^2}$$

Partielles Differenzieren (#2)

- Funktionen werden mit dem D-Operator partiell differenziert, wobei der Index der Funktionsvariable(n) übergeben wird, nach der abgeleitet werden soll:

```
> f := (x,y) -> cos(x*y)/y;
```

$$f := (x, y) \to \frac{\cos(x\,y)}{y}$$

```
> df1 := D[1](f);
```

$$df1 := (x, y) \to -\sin(x\,y)$$

Ableitung nach x: $\frac{\partial f}{\partial x}$

```
> d2f := D[2](f);
```

$$d2f := (x, y) \to -\frac{\sin(x\,y)\,x}{y} - \frac{\cos(x\,y)}{y^2}$$

Ableitung nach y: $\frac{\partial f}{\partial x}$

```
> d12f := D[1,2](f);
```

$$d12f := (x, y) \to -\cos(x\,y)\,x$$

Ableitung nach x und y: $\frac{\partial^2 f}{\partial x \partial y}$

- Mehrfache partielle Ableitungen mittels **$n** oder Wiederholungsoperator **@@**:

```
> D[2,1$2](f);
```

$$(x, y) \to \sin(x\,y)\,y\,x - \cos(x\,y)$$

```
> (D[1]@@2)(D[2](f));
```

$$(x, y) \to \sin(x\,y)\,y\,x - \cos(x\,y)$$

$$\frac{\partial^3 f}{\partial y \partial x^2}$$

Integrieren

Ableitung geht mechanisch, Integrieren ist eine Kunst

- Integrieren ist für Maple genauso ein schwieriges Problem, wie für Menschen.
- Maple ist mit vielen Integrationsansätzen vertraut:
  - Nachschlagen der Stammfunktion in einer Datenbank
  - Partielle Integration
  - Variablensubstitution
  - Partialbruchzerlegung
  - Spezielle Suchalgorithmen zur Bestimmung der Stammfunktion (Risch-Algorithmus)
- Erfolg ist für komplizierte Fälle keineswegs garantiert !
- Oft braucht Maple zusätzliche Hilfestellung (Annahme über Parameter im Integral)

Beim Integrieren muss der Benutzer mitdenken!

Unbestimmte Integrale

- Ausdrücke werden mit der Funktion **int()** integriert

```
> f := sin(x) * x;
```

$$f := \sin(x)\, x$$

```
> F := int(f, x);
```

$$F := \sin(x) - \cos(x)\, x$$

Integrationsvariable

Ausdruck, der integriert werden soll

- Im Gegensatz zur Ableitung gibt es keinen Integraloperator für Funktionen.
- Funktionen müssen zum Ausdruck konvertiert werden, und nach der Integration eventuell wieder in eine Funktionen verwandelt werden:

```
> f := x -> sin(x) * x;
```

$$f := x \to \sin(x)\, x$$

```
> int(f(x), x);
  F := unapply(%, x);
```

$$\sin(x) - \cos(x)\, x$$

$$F := x \to \sin(x) - \cos(x)\, x$$

Träge Integration

- Auch die Integration besitzt eine „träge“ Form **Int()**.
  Auswertung kann mittels **value()** erreicht werden:

```
> f := sin(x) * x;
```

$$f := \sin(x)\, x$$

```
> Int(f,x);
  F := value(%);
```

$$\int \sin(x)\, x \, dx$$

$$F := \sin(x) - \cos(x)\, x$$

- Die träge Form kann dazu verwendet werden, anzuzeigen, was genau integriert wird.

- Mit Hilfe der **infolevel[int]** Variable können interne Details der Integration angezeigt werden:

```
> infolevel[integrate] := 2;
```

Menge der angezeigten Information

$$infolevel_{int} := 2$$

```
> int(sin(x)*x, x);
`int/indef1:`, `first-stage indefinite integration`
`int/indef2:`, `second-stage indefinite integration`
`int/trigon:`, `case of integrand containing trigs`
```

$$\sin(x) - \cos(x)\, x$$

- Wenn eine Stammfunktion nicht gefunden werden kann, wird das Integral erneut ausgegeben:

```
> int(sin(x)/ln(x), x);
```

$$\int \frac{\sin(x)}{\ln(x)}\, dx$$

Bestimmte Integrale

- Bei bestimmten Integralen wird der Integrationsbereich als Parameter übergeben:

```
> f := tan(x);
```

$$f := \tan(x)$$

```
> Int(tan(x), x=0..1);
  res := value(%);
```

Integrationsbereich

$$\int_0^1 \tan(x)\, dx$$

$$res := -\ln(\cos(1))$$

- Das Ergebnis kann mittels **evalf()** numerisch ausgewertet werden:

```
> evalf(res);
```

$$0.6156264703$$

- Wird nur eine numerische Auswertung gewünscht (z.B. weil analytische Stammfunktion nicht bekannt ist), sollte die träge Form der Integration mit **evalf()** kombiniert werden:

```
> Int(sin(x)/ln(x), x=2..3);
  evalf(%);
```

$$\int_2^3 \frac{\sin(x)}{\ln(x)}\, dx$$

$$0.6743806034$$

- Maple kann auch mit uneigentlichen Integralen umgehen:

```
> Int(1/x, x=1..infinity);
  value(%);
```

$$\int_{1}^{\infty} \frac{1}{x}\, dx$$

$$\infty$$

Uneigentliches Integral mit Unendlich als Ergebnis

```
> Int(1/x^2, x=1..infinity);
  value(%);
```

$$\int_{1}^{\infty} \frac{1}{x^{2}}\, dx$$

$$1$$

Uneigentliches Integral mit endlichem Ergebnis

Mehrfachintegrale

- Mehrfachintegrale werden durch geschachtelte **int()** bzw. **Int()** Funktionen eingegeben:

```
> f := x^2 * y;
```

$$f := x^2 y$$

```
> Int(Int(f, x), y);
  value(%);
```

$$\int\int x^2 y \, dx \, dy$$

$$\frac{1}{6} x^3 y^2$$

- Numerische Mehrfachintegrale analog:

```
> Int(Int(f, x=-1..1), y=-1..1);
  evalf(%);
```

$$\int_{-1}^{1}\int_{-1}^{1} x^2 y \, dx \, dy$$

$$-1.033485533\ 10^{-18}$$

ungleich Null wegen endlicher Darstellung in der Fließkommaarithmetik (mit Digits := 20 wäre das Null)

Hilfestellung beim Integrieren

- Machmal braucht Maple Hilfe, um ein Integral berechnen/vereinfachen zu können:

```
> Int(exp(-a * x), x=0..infinity);
  value(%);
```

$$\int_0^{\infty} e^{(-a\,x)}\,dx$$

$$\lim_{x \to \infty} \left(-\frac{e^{(-a\,x)} - 1}{a} \right)$$

nicht auswertbar, da Vorzeichen von *a* unbekannt

```
> Int(exp(-a * x), x=0..infinity);
  value(%) assuming(a > 0);
```

Annahme über a

$$\int_0^{\infty} e^{(-a\,x)}\,dx$$

$$\frac{1}{a}$$

Annahme muss bei der Auswertung gestellt werden!

Hilfestellung beim Integrieren (#2)

- Oft wird durch Wechsel des Koordinatensystems die Integration erleichtert/beschleunigt:

```
> Int(Int(Int(1, x=-sqrt(R^2-y^2-z^2)..sqrt(R^2-y^2-z^2)),
  y=-sqrt(R^2-z^2)..sqrt(R^2-z^2)), z=-R..R);
  simplify(value(%));
```

$$\int_{-R}^{R} \int_{-\sqrt{R^2 - z^2}}^{\sqrt{R^2 - z^2}} \int_{-\sqrt{R^2 - y^2 - z^2}}^{\sqrt{R^2 - y^2 - z^2}} 1 \, dx \, dy \, dz$$

$$\frac{4}{3} R^3 \pi$$

Volumen einer Kugel in kartesischen Koordinaten

```
> Int(Int(Int(r^2*sin(theta), r=0..R), theta=0..Pi),
  phi=0..2*Pi);
  simplify(value(%));
```

$$\int_{0}^{2\pi} \int_{0}^{\pi} \int_{0}^{R} r^2 \sin(\theta) \, dr \, d\theta \, d\phi$$

$$\frac{4}{3} R^3 \pi$$

Volumen der Kugel in Kugelkoordinaten

ca. 400fach schneller!

- Bei der Taylorreihe wird eine beliebige Funktion um einen ausgewählten Punkt x_0 mit einem Polynom genähert:

$$p_n(x)=\sum_{i=0}^{n}\frac{1}{i!}\cdot f^{(i)}(x_0)\cdot(x-x_0)^i$$

Näherungspolynom n-ten Grades

i-te Ableitung von f ($f^{(0)}=f$)



Taylorreihe (#2)

- In Maple kann das Taylorpolynom mit dem Befehl **taylor()** berechnet werden:

```
> t := taylor(exp(x), x=0, 4);
```

$$t := 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3 + O(x^4)$$

Anzahl der Terme (Grad des Polynoms + 1)

Funktion

Variable und Entwicklungspunkt

Niedrigste Ordnung der vernachlässigtenTerme

- Terme mit Nullkoeffizienten werden nicht ausgegeben:

```
> t := taylor(sin(x), x=0, 4);
```

$$t := x - \frac{1}{6}x^3 + O(x^4)$$

- Die Taylorreihe kan mit **convert()** zu einem auswertbaren Polynom konvertiert werden:

```
> t_poly := convert(t, polynom);
```

$$t\_poly := 1 + x + \frac{1}{2}x^2 + \frac{1}{6}x^3$$

Maplepacket Student

- Der Maplepacket **Student** ermöglicht Integration und Differentiation in Teilschritten, mit der gezielten Anwendung ausgewählter Regel:

Anwendung der Produktregel

Anwendung der Potenzregel

Ableiten von sin

Anzeigen der einzelnen Schritte

```
> with(Student[Calculus1]):
> f := Diff(x^2*sin(x), x);
```

$$f := \frac{d}{dx}(x^2 \sin(x))$$

```
> Rule[product](f);
```

$$\frac{d}{dx}(x^2 \sin(x)) = \left(\frac{d}{dx}(x^2)\right)\sin(x) + x^2\left(\frac{d}{dx}\sin(x)\right)$$

```
> Rule[`^`](%);
```

$$\frac{d}{dx}(x^2 \sin(x)) = 2\,x \sin(x) + x^2\left(\frac{d}{dx}\sin(x)\right)$$

```
> Rule[sin](%);
```

$$\frac{d}{dx}(x^2 \sin(x)) = 2\,x \sin(x) + x^2 \cos(x)$$

```
> ShowSteps();
```

$$\begin{aligned}\frac{d}{dx}(x^2 \sin(x)) &= \left(\frac{d}{dx}(x^2)\right)\sin(x) + x^2\left(\frac{d}{dx}\sin(x)\right)\\ &= 2\,x \sin(x) + x^2\left(\frac{d}{dx}\sin(x)\right)\\ &= 2\,x \sin(x) + x^2 \cos(x)\end{aligned}$$

Maplepacket Student (#2)

- Besitzt interaktive Anwendungen zum Differenzieren (**DiffTutor()**), Integrieren (**IntTutor()**) und zur Taylornäherung (**TaylorApproximationTutor()**)



- Für weitere Infos siehe Hilfe in Maple (**?Student[Calculus1]**)

Abbildungen als Variablen

- Abbildungen können in Variablen gespeichert werden und mit dem Befehl **display()** (benötigt das Packet **plots**!) wieder angezeigt werden:

```
> with(plots):
Warning, the name changecoords has been redefinec
> pl1 := plot(sin(x), x=-2*Pi..2*Pi, y=-1..1):
> display(pl1);
```



- Bei der Variablendeklaration sollte die Ausgabe mit „**:**“ unterdrückt werden, ansonsten viel unnötige Information:

```
> pl1 := plot(sin(x), x=-2*Pi..2*Pi, y=-1..1);
```

pl1 := *INTERFACE_PLOT*(*CURVES*([

$[-6.28318529461999997, 1.25595865048264204\ 10^{-8}],$

$[-6.14622973609543522, 0.136527830408832623],$

$[-6.00927417757086957, 0.270498819883406905],$

Abbildungen als Variablen (#2)

- Display kann mehrere Abbildungen gleichzeitig anzeigen-
- Besonders praktisch, wenn verschiedene Darstellungsmoden in einer Abbildung erscheinen sollen:

```
> pl1 := plot(sin(x), x=-2*Pi..2*Pi, y=-1..1):
> pl2 := plot([[0, sin(0)], [Pi/2, sin(Pi/2)]],
  style=point, symbolsize=20, color=blue):
> display({pl1,pl2});
```

normale Funktionendarstellung

Darstellung von Punkten
(Punkte als Liste von [x,y] Paaren



- Wenn die Option **insequence=true** angegeben wird, werden die Abbildungen nacheinander als Animation angezeigt:
  (Das Bild muss angeklickt werden, damit das Menu zum Starten der Animation erscheint!).

```
> pl1 := seq(plot(convert(taylor(sin(x), x=0, i),
  polynom), x=-2*Pi..2*Pi, y=-1..1), i=0..12):
> display(pl1, insequence=true);
```



- Weitere Animationsmöglichkeit bietet die **animate()** Funktion. Siehe Maple-Hilfe.

Vielen Dank für die Aufmerksamkeit!

Jetzt sind Sie daran!